
1 Introduction 

1.1 Summary 

SAP Solution Manager IT Service Management is using Substatus or Status Reason to 
give end-users the possibility to specify why the current status is choosen. This is a very 
good feature which can be used for initial status (Why transaction is created?) final 
status (Why transaction is closed/withdrawn?), approval and confirmation status (Why it 
is approved/confirmed/rejected?). The status „Sucessfully Tested“ or „Back to 
Development“ is also a confirmation status in this meaning. 

We don’t want to discuss here, why SAP SE introduced a new concept of substatus / 
status reason which was already existing in SAP CRM long time ago and which is 
heavily used by leads, opportunities, … This is another topic which is out of scope of 
this document. 

In this document I want to tell you how to enable this feature for Change Request 
Management, because it is helpful for ChaRM as well. 

1.2 Quick Guide 

 Configuration of substatus can be done like for ITSM. No additional customizing table 
is needed. 

 Substatus is available as field /AICRM/REASON_ID of table CRMD_SERVICE_H. 
This is true for ITSM and ChaRM. No field enhancement is needed. If we want to 
activate processing/change logging for this field we just have to configure it using 
these technical information. 

 Substatus is available as field //BTADMINH/EXT./AICRM/REASON_ID in UI 
configuration of details form. This is true for ITSM and ChaRM (CR and CD). We just 
need to make the field visible. However ChaRM is not supporting any value help or 
input readiness check. This needs to be implemented (see below). 

 Substatus is available as Search Criteria and Search Result List Field. This is true for 
ITSM and ChaRM (CR and CD). We just need to make the field visible. However 
ITSM and ChaRM are not supporting any value help (search criteria) and key to text 
conversion (result list field). This needs to be implemented (see below). 

 We still have the issue, that the status change is performed by ppf actions in ChaRM. 
Therefore we have no chance to maintain the substatus / status reason /closure code 
on change to a final status, because the transaction is final after status change. Only 
a popup solution can help here which will need a medium-size development (see 
below). 



 
 

ChaRM Substatus | Peter Weigel Page 4 of 63 

www.hybrid-eichhoernchen.de

1.3 Original Requirement of customer 

When withdrawing SolMan test tasks (for any environment / phase), the possible 
reason codes should be as follows: 

 Created by mistake 

 Tested elsewhere 

 Effort outweighs Risk 

 Function not used 

  

When closing test tasks as “Failed but ok to go”, the possible reason codes should 
be as follows: 

 Test result differs from Expectation 

 Not tested: lack of resources 

 Not tested: environment issue 

 Not tested: lack of resources 

  

Reporting:  

 Assumption is that reporting against SolMan test task status (Successful, Withdrawn, 
Failed but ok to go) is already possible. Contradicting the original ALM requirement, 
no reporting on substatus values are needed yet. Neither Solman Reports nor 
BW/BO reports. 

 

Withdrawn of Change Request, Manual Activities and Build CDs: 

 We have a lot of status changes where substatus or multi line texts needs to be 
maintained. 
This popup solution makes sense here as well to guide the end-users. 

 Currently there is no requirement to activate it. 

 The technical solution should be flexible to support all these use cases as well (if we 
will configure it in future). 

1.4 Technical Ideas 

Technical ideas (field): 

 Multi-Level Categorization 

 ITSM Status Reason (Substatus) 

 SAP CRM Status Reason 

 New Field using AET 

Solution 1: 

 Multi Level: 1 - Close / Withdrawn, 2 - Code (see above) 

 Auto trigger of status change (category as start condition) 

 --> too dangerous because of unexpected automatism 

  

Solution 2: 

 Multi Level: 1 - Close / Withdrawn, 2 - Code (see above) 

 Validation on status change (consistency check) 

 --> Additional customizing table or mapping to target status based on MLC ID naming 

 Pro: 

 Effort is low (3 PD) 

 Use of existing ChaRM framework/features 

 Contra: 

 MLC values are not filtered based on status (End-Users gets confused 
because mismatch of MLC and status change action) 

  

Solution 3: 

 Status reason (ITSM or SAP CRM) 

 Popup for status reason (before or after action) 

 Pro: 

 end-user friendly 

 Contra: 

 More complex development logic 

 prototype necessary 

 Effort high (8 PD) 

  

Question ITSM or SAP CRM status reason: 

 What is planned in future? 

 Reporting possible out of the box? 

 --> ITSM Status Reason is easier to implement 

1.5 Version History 

Version Author Date Comment 

1 Peter Weigel 27.10.2016 First "Quick and Dirty" Version (may 
contain errors in content and layout) 

2 Peter Weigel 10.11.2016 Status Change Cancel Feature 

1.6 Literature, Disclaimer, Contact and Download 

Literature 

This document is based on information from SAP Online Library, Implementation Guide 
of SAP Solution Manager 7.1, several SAP Notes and several SCN articles. These 
piece of information were enriched by the authors knowledge and experience. 

Disclaimer 

http://www.hybrid-eichhoernchen.de/rechtliche-hinweise/ 

Contact 

Peter Weigel 
Hyazinthenstr. 6 
D-06122 Halle / Saale 
Phone: +49 170 5337567 
E-Mail: peter.weigel@hybrid-eichhoernchen.de 
Web: www.hybrid-eichhoernchen.de 
 
Download 

You are allowed to download the latest version of this document for free: www.hybrid-
eichhoernchen.de. 

http://www.hybrid-eichhoernchen.de/rechtliche-hinweise/
http://www.hybrid-eichhoernchen.de/rechtliche-hinweise/
mailto:peter.weigel@hybrid-eichhoernchen.de
http://www.hybrid-eichhoernchen.de/
http://www.hybrid-eichhoernchen.de/
http://www.hybrid-eichhoernchen.de/


 
 

ChaRM Substatus | Peter Weigel Page 7 of 63 

www.hybrid-eichhoernchen.de

1 ITSM Substatus 

1.1 Customizing: Specify Substatus Values 

We are using ITSM Substatus. The corresponding configuration can be found in the 
ITSM section. The requested "closure codes" for test tasks are now maintained there. 
However the description is technically restricted to 30 characters only. Technically the 
Substatus keys don't need to be unique. That means the same key can be re-used for 
different user status resp. process types. However the description can be maintained 
only once, therefore the technical and functional meaning should be the same. 

1.2 UI Configuration: Display Substatus 

The Substatus is stored in field /AICRM/REASON_ID of table CRMD_SERVICE_H. In 
UI it is available as attribute //BTADMINH/EXT./AICRM/REASON_ID (with the name 
"Substatus"). It is available for Incident, Change Request and Change Documents 
(including Test Tasks and Manual Activities). It is now configured for ZTT1 as visible 
field in details form next to "Status". 

The value help and input readiness check is only implemented in Incident solution 
(AIC_INCIDENT_H). The context node AIC_INCIDENT_H/IncidentHeaderEF-
>BTADMINH is using controller class CL_AIC_INCI_INCIDENTHEADE_CN03 while 
AIC_CMCR_H/AICCMCRHeaderEF->BTADMINH is using 
CL_SRQM_H_BTADMINH_CN. Because CL_AIC_INCI_INCIDENTHEADE_CN03 is 
inheriting from CL_SRQM_H_BTADMINH_CN we just need to adjust the inheriting 
hierarchy of the CR context node BTAdminH. There is no need to copy source code. 
However, this is a SAP bug. 

 
 

Before Adjustment: 

After Adjustment: 

 
 

Still valid: 

 
 

Controller class CL_AIC_INCI_INCIDENTHEADE_CN03 has knowledge about the view 
controller. The corresponding assignment will fail now, because the view controller class 
of CR is different from INC. However, the assignment is currently not needed for CR. 
Therefore we just need to ensure that we will get no short dump when trying to take 
over view controller reference. 

Same solution for AIC_CMCD_H: 

Before: 

 
 

After: 

Caution: I am very sure, that SAP will change the inheritance hierarchy for 
AIC_CMCR_H and AIC_CMCD_H in future and replace CL_SRQM_H_BTADMIN_H by 
another class. In this case we have to adjust the inheritance again! However, we have 
to do it anyway independent from this adjustment. In worst case (if we cannot use 
inheritance) we have to copy the source code of attribute "Substatus" from 
AIC_INCIDENT_H after upgrade. 

1.3 UI Configuration: Search for Substatus 

The Substatus is available for Incident, Change Request and Change Documents 
(including Test Tasks and Manual Activities). It is now configured for Change 
Documents (incl. Test Tasks) as available and visible search criteria and search result 
field. In SAP standard only operator "Is" is supported. We improved it to support some 
more operators (IS and IS NOT). 

SAP standard is not providing any value help for the "substatus" search criteria. 
Because we want to have a value help, we needed to implement it as enhancement. 

CD Search Criteria: 

METHOD get_v_/aicrm/reason_id . 

 

  DATA: 

    lt_list_proc_type  TYPE crmt_process_type_tab, 

    lt_range_proc_type TYPE RANGE OF crmt_process_type, 

    ls_range_proc_type LIKE LINE OF lt_range_proc_type. 

 

  FIELD-SYMBOLS: 

    <fv_proc_type> LIKE LINE OF lt_list_proc_type. 

 

 

* Get all relevant process types (see value help getter for process_type) 

  me->get_all_proc_type( IMPORTING et_proc_type = lt_list_proc_type  ). 

 

*Build select option. 

  LOOP AT lt_list_proc_type ASSIGNING <fv_proc_type>. 

    ls_range_proc_type-sign = 'I'. 

    ls_range_proc_type-option = 'EQ'. 

    ls_range_proc_type-low = <fv_proc_type>. 

    APPEND ls_range_proc_type TO lt_range_proc_type. 

  ENDLOOP. 

 

*Get all possible entries. 

  SELECT DISTINCT reason_id AS key txt30 AS value 

    INTO CORRESPONDING FIELDS OF TABLE cs_result-ddlb_options 

    FROM aic_stat_reasont 

   WHERE spras = sy-langu AND 

         reason_id IN ( SELECT reason_id 

                          FROM aic_stat_reason 

                         WHERE user_stat_proc IN ( SELECT user_stat_proc 

                                                     FROM crmc_proc_type 

                                                    WHERE process_type IN lt_range_proc_type ) ) 

   ORDER BY reason_id. 

 

*Add empty line. 

  APPEND INITIAL LINE TO cs_result-ddlb_options. 

 

ENDMETHOD. 

 

CD Search Result: 

METHOD get_/aicrm/reason_id . 

 

  DATA: lr_current TYPE REF TO  if_bol_bo_property_access, 

        lv_key     TYPE         aic_reasonid. 

 

*Get entity. 

  IF iterator IS BOUND. 

    lr_current = iterator->get_current( ). 

  ELSE. 

    lr_current = collection_wrapper->get_current( ). 

  ENDIF. 

 

*Get key. 

  lv_key = lr_current->get_property_as_string( iv_attr_name = '/AICRM/REASON_ID' ). 

 

*use key as value if text cannot be found. 

  value = lv_key. 

 

*Try to find text. 

  SELECT SINGLE txt30 

    INTO value 

    FROM aic_stat_reasont 

    WHERE reason_id = lv_key AND 

          spras = sy-langu. 

 

ENDMETHOD. 

CR Search Criteria. 

And CR Search Result: 

Please note: Method cl_ai_crm_im_utility=>get_reason_ids cannot be used because 
this is only working for a specific transaction guid. 

1.4 Change Logs 

We want to log all changes to the Sub Status as Change Log. This feature is now 
activated for all existing transaction types. 

2 Multi Line Texts 

Together with substatus we want to maintain a multi-line text (i.e. Withdrawal Reason or 
Test Failure Reason) on status change. Because text type can be different, this text 
type needs to be assigned to a specific status transition. Additionally this text type 
needs to be accessible via BOL/GenIL. 

 

2.1 Customizing: Assign Text Type to Status Transition 

This maintenance is customer specific. 

 

Maintenance View ???_CSS_S_V and customizing table ???_CSS_S_C are extending 
the already existing table ???_STAT by the text type field. 

2.2 Enable BOL/GenIL access for Text Type 

3 Sub Status Popup 

3.1 Customizing: Assign Dialog Box to a PPF Action 

We want to show a popup before executing a status change action. The popup is used 
to select a sub status and to maintain a multi-line text. 

The popup cannot be show after status change because the selection/maintenance 
done in the popup might be needed for some consistency checks. The popup cannot be 
shown after status change because the transaction is not changeable anymore after 
change to a final status. 

On status change some consistency checks will be executed which might cancel the 
status change. One of these consistency check is a check for a valid sub status. In case 
of status change cancel, the already selected sub status will be removed. These both 
features (consistency check and sub status clean-up) are not part of this section. 
However we might need to keep it in mind to understand the whole solution. 

 

 

 
 

To activate the popup we need to configure it including ui component, BSP interface, 
BSP UI usage, OTR Alias. We also need to set the flag "Before PPF Action" and 

"Active". Actually the popup solution can be configured for all Change Requests and 
Change Documents. It cannot be configured for Incidents, problems, Service Requests, 
because the Popup framework is not supported here. 

3.2 Text: Popup Title 

The OTR Alias for the popup title is defined it transaction SOTR_EDIT: 

 

3.3 UI Component CSS_UI 

The popup is implemented as own UI component CSS_UI. 

 

 

3.4 Component Controller 

The ui component has a component controller with context node to hold popup data 
(POPDATA) like action name and transaction root (BTAdminH). Context node of 
BTAdminH is inheriting from AIC_INCIDENT_H. It is using all implementations already 
done for Incident but it can be enhanced. Context Node POPDATA is re-using 

implementation of Popup Example (AIC_POPUPEXAMP) - it cannot be enhanced 
actually. 

 

 

 

 

3.5 Window CSS_UI/MainWindow 

The root and entry point of the popup is the window which is showing the popup view. 

Context node POPDATA is connected with the corresponding context nodes of the 
component controller. POPDATA is the standard implementation of 
AIC_POPUPEXAMP (like component controller).  

The plugs are described in section 3.9 and 3.10. 

3.6 View CSS_UI/SubstatusEF 

View CSS_UI/SubstatusEF is embedded in Window CSS_UI/MainWindow. It will be 
shown as soon as the Window will be shown. 

Context Nodes POPDATA and BTADMINH are connected with the corresponding 
context nodes of the component controller (not visible in screenshot). Context nodes 
BTTEXTSETH and BTTEXT are dependent on context nodes BTAdminH. Note: All 
changes done in BTADMINH, BTTEXTSETH and BTTEXT are real changes and will be 
send to BOL/GenIL and API automatically. 

POPDATA is the standard implementation of AIC_POPUPEXAMP (like component 
controller and window).  

BTADMINH is again inheriting the standard implementation from AIC_INCIDENT_H 
(which is supporting Sub Status, like component controller). However some 
adjustements wetre needed to show the target user status and to allow selection of 
substatus of the target user status. More details concerning this topic can be found in 
section 3.7. 

BTTEXTSETH is generated by wizard based on relation BTAdminH-> 
BTHeaderTextSet. BTTEXT is generated by wizard based on relation BTTEXTSETH-> 
BTTextH_<TDID>. However after generation the implementation was adjusted to be 
more dynamically. More details concerning this topic can be found in section 3.8. 

Actually there is only one UI configuration where user status is read-only and substatus 
and text type are mandatory. It can happen that we need to change this in future in case 
sub status or text type are not mandatory in some cases. 

This view is not using view group contexts because it is always in edit mode. 

Note: Actually the button texts "Perform Status Change" and "Cancel" are hard coded. If 
we will need a multi-lingual solution in future, we have to adjust this. Please have a look 
at the Transport creation popup to see how to do this. 

3.7 Context Node BTAdminH 

As mentioned context node BTAdminH is inheriting all logic from AIC_INCIDENT_H. 
This means user status and substatus are already supported for the current user status. 
But we want to show the target user status. We want to select a substatus for the target 
user status. 

Context Node BTAdminH needs to remember his owner to have access to other context 
nodes of the same context node controller.  That is why we need to save the reference 
on init. 

 

 

Get Target User Status (GET_LCSTATUS): 

 Call super method to get current user status. 

 Get AdminH and extract process type. 

 Get POPDATA and extract action name. 

 Determine target user status from table ???_STAT. 

METHOD get_lcstatus. 

 

*Get Current User Status. 

  CALL METHOD super->get_lcstatus 

    EXPORTING 

      attribute_path = attribute_path 

      iterator       = iterator 

    RECEIVING 

      value          = value. 

 

*------------------------------------------------------------------------- 

 

  DATA: 

    lr_adminh      TYPE REF TO cl_crm_bol_entity, 

    lr_popdata     TYPE REF TO cl_bsp_wd_value_node, 

    lv_proc_type   TYPE crmt_process_type, 

    lv_action_name TYPE ppfdtt. 

 

*Get BTAdminH. 

  IF iterator IS BOUND. 

    lr_adminh ?= iterator->get_current( ). 

  ELSE. 

    lr_adminh ?= collection_wrapper->get_current( ). 

  ENDIF. 

 

*Extract Process Type. 

  lr_adminh-

>get_property_as_value( EXPORTING iv_attr_name = 'PROCESS_TYPE' IMPORTING ev_result = lv_proc_typ

e ). 

 

*Get Popup Data. 

  lr_popdata ?= me->???_owner->typed_context->popdata->collection_wrapper->get_current( ). 

 

*Extract Action Name. 

  lr_popdata-

>get_property_as_value( EXPORTING iv_attr_name = 'PPF_ACTION' IMPORTING ev_result = lv_action_nam

e ). 

 

*Determine Target Status. 

*Option 1: Get PPF Action and extract parameter USER_STATUS from PPF Container. Look at Popup Exa

mple BADI (class CL_EX_AIC_POPUP_EXAMPLE) to find out how to do that. 

*Option 2: Get target status from customizing table /SALM/PRJ_STAT. 

 

  SELECT SINGLE user_status_new 

    INTO value 

    FROM ???_stat 

    WHERE process_type = lv_proc_type AND 

          user_status = value AND 

          action = lv_action_name. 

 

ENDMETHOD. 

 

Get Substatus value help for target user status (GET_V_/AICRM/REASON_ID): 

 Check whether substatus values are already determined. if not: 

 Get BTAdminH 

 Extract transaction guid 

 Get target user status using method GET_LCSTATUS 

 Get Substatus values using method cl_ai_crm_im_utility=>get_reason_ids 

 Build value help descriptor. 

 

METHOD get_v_/aicrm/reason_id. 

*CALL METHOD SUPER->GET_V_/AICRM/REASON_ID 

**  EXPORTING 

**    iv_mode                 = IF_BSP_WD_MODEL_SETTER_GETTER=>RUNTIME_MODE 

**    iv_index                = 

*  RECEIVING 

*    RV_VALUEHELP_DESCRIPTOR = 

*    . 

 

  DATA: lt_ddlb           TYPE         bsp_wd_dropdown_table, 

        ls_ddlb           TYPE         bsp_wd_dropdown_line, 

        cr_ddlb           TYPE REF TO  cl_crm_uiu_ddlb, 

        lr_btadminh       TYPE REF TO  cl_crm_bol_entity, 

        lr_current        TYPE REF TO  if_bol_bo_property_access. 

 

  DATA:  lv_reason_id     TYPE aic_reasonid, 

         lv_guid          TYPE crmt_object_guid, 

         lt_stat_reason   TYPE aic_stat_reason_tt, 

         ls_stat_reason   TYPE aic_s_stat_reason, 

         lv_dummy_key     TYPE char40. 

 

  IF me->mt_stat_reason IS INITIAL. 

    lr_current = collection_wrapper->get_current( ). 

    CHECK lr_current IS BOUND. 

    lr_btadminh ?= lr_current. 

* GET PARENT ENTITY 

    TRY. 

        WHILE lr_btadminh->get_name( ) NE 'BTAdminH'. 

 

          lr_btadminh = lr_btadminh->get_parent( ). 

 

        ENDWHILE. 

      CATCH cx_sy_ref_is_initial. 

    ENDTRY. 

* GET PROCESSISNG ORDER GUID 

    IF lr_btadminh IS BOUND. 

 

      lr_btadminh->get_property_as_value( EXPORTING iv_attr_name = 'GUID' 

                                          IMPORTING ev_result    = lv_guid ). 

 

*------------------------------------------------------------------------- 

*Get Target User Status and Reason ID for it. 

 

      DATA: 

        lv_status TYPE j_estat. 

 

      lv_status = me->get_lcstatus( attribute_path = '' ). 

 

      cl_ai_crm_im_utility=>get_reason_ids( EXPORTING iv_guid = lv_guid iv_status = lv_status 

                                            IMPORTING et_stat_reason = mt_stat_reason ). 

 

*------------------------------------------------------------------------- 

 

*      cl_ai_crm_im_utility=>get_reason_ids( EXPORTING iv_guid = lv_guid 

*                                            IMPORTING et_stat_reason = mt_stat_reason ). 

 

    ENDIF. 

  ENDIF. 

 

  lv_dummy_key = lv_reason_id. 

 

  IF gr_ddlb_reason IS NOT BOUND. 

    CREATE OBJECT gr_ddlb_reason 

      EXPORTING 

        iv_source_type = 'T'. 

  ENDIF. 

*   add initial line 

  ls_ddlb-key = lv_dummy_key. 

  APPEND ls_ddlb TO lt_ddlb. 

 

 

  LOOP AT mt_stat_reason INTO ls_stat_reason. 

 

 

    ls_ddlb-key   =   ls_stat_reason-reason_id. 

    ls_ddlb-value =   ls_stat_reason-txt30. 

 

    APPEND ls_ddlb TO lt_ddlb. 

 

  ENDLOOP. 

 

  gr_ddlb_reason->set_selection_table( lt_ddlb ). 

 

  rv_valuehelp_descriptor = gr_ddlb_reason. 

 

 

ENDMETHOD. 

 

Get Substatus values (cl_ai_crm_im_utility=>get_reason_ids): 

In SAP standard, method cl_ai_crm_im_utility=>get_reason_ids is not able to determine 
substatus values for a given user status. Therefore this method needed an 
enhancement. 

METHOD get_reason_ids. 

 

  DATA: ls_status       TYPE crmt_status_wrk, 

        ls_stat_reason  like line of gt_reason, 

        ls_reason       TYPE aic_s_stat_reason. 

 

clear: et_stat_reason. 

 

  CALL FUNCTION 'CRM_STATUS_READ_OW' 

    EXPORTING 

      iv_guid                = iv_guid 

    IMPORTING 

      es_current_user_status = ls_status 

    EXCEPTIONS 

      not_found              = 1 

      OTHERS                 = 2. 

 

IF gt_reason IS INITIAL. 

 

    SELECT * FROM aic_stat_reason INTO CORRESPONDING FIELDS OF ls_stat_reason.  "#EC CI_NOWHERE 

      IF sy-subrc NE 0. 

        EXIT. 

      ENDIF. 

      SELECT spras txt30 FROM aic_stat_reasont INTO CORRESPONDING FIELDS OF ls_stat_reason 

       WHERE reason_id EQ ls_stat_reason-reason_id. 

        APPEND ls_stat_reason TO gt_reason. 

      ENDSELECT. 

    ENDSELECT. 

  ENDIF. 

 

  LOOP AT gt_reason INTO ls_stat_reason WHERE user_stat_proc = ls_status-user_stat_proc 

                                                  AND status = ls_status-status 

                                                  AND spras  = sy-langu. 

    MOVE-CORRESPONDING ls_stat_reason TO ls_reason. 

    APPEND ls_reason TO et_stat_reason. 

  ENDLOOP. 

 

 

 

 

 

"""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""$"$\

SE:(1) Class CL_AI_CRM_IM_UTILITY, Method GET_REASON_IDS, End                                    

                                                        A 

*$*$-Start: (1)---------------------------------------------------------------------------------

$*$* 

ENHANCEMENT 1  ???_CSS_CRM_GET_REASON_ID.    "active version 

 

*Get Status Reason for a specific status value. 

    IF IV_STATUS IS NOT INITIAL. 

 

      CLEAR et_stat_reason. 

      LOOP AT gt_reason INTO ls_stat_reason WHERE user_stat_proc = ls_status-user_stat_proc 

                                                  AND status = iv_status 

                                                  AND spras  = sy-langu. 

        MOVE-CORRESPONDING ls_stat_reason TO ls_reason. 

        APPEND ls_reason TO et_stat_reason. 

      ENDLOOP. 

 

    ENDIF. 

ENDENHANCEMENT. 

*$*$-End:   (1)---------------------------------------------------------------------------------

$*$* 

ENDMETHOD. 

 

Check Input Readiness (GET_I_/AICRM/REASON_ID) 

Analogous to GET_V_/AICRM/REASON_ID we copied the sap standard source code 
and replaced the current user status by the target user status. 

The sap standard solution contains a very dirty code to clear the substatus in case no 
valid values are available. This solution is dirty because the input readiness check is not 
designed to do such things. That is why we deactivated this solution. However it is still 
active and working in details form of the change transaction. 

METHOD get_i_/aicrm/reason_id. 

*CALL METHOD SUPER->GET_I_/AICRM/REASON_ID 

**  EXPORTING 

**    iterator    = 

*  RECEIVING 

*    RV_DISABLED = 

*    . 

 

  DATA: lr_current       TYPE REF TO if_bol_bo_property_access, 

        lr_btadminh      TYPE REF TO cl_crm_bol_entity, 

        lr_entity        TYPE REF TO cl_crm_bol_entity, 

        lr_coll          TYPE REF TO if_bol_entity_col, 

        lt_stat_reason   TYPE aic_stat_reason_tt, 

        lv_guid          TYPE crmt_object_guid. 

 

  rv_disabled = 'TRUE'. 

  IF iterator IS BOUND. 

    lr_current = iterator->get_current( ). 

  ELSE. 

    lr_current = collection_wrapper->get_current( ). 

  ENDIF. 

 

  CHECK lr_current IS BOUND. 

  lr_btadminh ?= lr_current. 

* GET PARENT ENTITY 

  TRY. 

      WHILE lr_btadminh->get_name( ) NE 'BTAdminH'. 

 

        lr_btadminh = lr_btadminh->get_parent( ). 

 

      ENDWHILE. 

    CATCH cx_sy_ref_is_initial. 

  ENDTRY. 

* GET PROCESSISNG ORDER GUID 

 

  IF lr_btadminh IS BOUND. 

 

    lr_btadminh->get_property_as_value( EXPORTING iv_attr_name = 'GUID' 

                                        IMPORTING ev_result    = lv_guid ). 

 

*------------------------------------------------------------------------- 

*Get Target User Status and Reason ID for it. 

 

    DATA: 

      lv_status TYPE j_estat. 

 

    lv_status = me->get_lcstatus( attribute_path = '' ). 

 

    cl_ai_crm_im_utility=>get_reason_ids( EXPORTING iv_guid = lv_guid iv_status = lv_status 

                                          IMPORTING et_stat_reason = mt_stat_reason ). 

 

*------------------------------------------------------------------------- 

 

*    cl_ai_crm_im_utility=>get_reason_ids( EXPORTING iv_guid = lv_guid 

*                                          IMPORTING et_stat_reason = mt_stat_reason ). 

 

  ENDIF. 

 

  IF mt_stat_reason IS NOT INITIAL. 

 

    rv_disabled = 'FALSE'. 

  ELSE. 

 

*    TRY. 

* 

*        lr_entity ?= lr_current. 

*        lr_coll = lr_entity-

>get_related_entities( iv_relation_name = 'BTHeaderServiceExt' ). "#EC NOTEXT 

* 

*        lr_current = lr_coll->get_current( ). 

*        IF lr_current IS NOT BOUND. 

*          RETURN. 

*        ENDIF. 

* 

*        TRY. 

*            lr_current->set_property_as_string( iv_attr_name = '/AICRM/REASON_ID' 

*                                                iv_value     = space ). 

*          CATCH cx_crm_cic_parameter_error. 

*        ENDTRY. 

* 

*      CATCH cx_sy_ref_is_initial cx_sy_move_cast_error 

*            cx_crm_genil_model_error. 

*        RETURN. 

*    ENDTRY. 

 

 

 

  ENDIF. 

 

 

 

 

 

*    rv_disabled = 'TRUE'. 

*    if iterator is bound. 

*      current = iterator->get_current( ). 

*    else. 

*      current = collection_wrapper->get_current( ). 

*    endif. 

* 

*  TRY. 

* 

*        IF current->is_property_readonly( 

*                      '/AICRM/REASON_ID' ) = abap_false. "#EC NOTEXT 

*          rv_disabled = 'FALSE'. 

*        ENDIF. 

* 

*    CATCH cx_sy_ref_is_initial cx_sy_move_cast_error 

*          cx_crm_genil_model_error. 

*      RETURN. 

*  ENDTRY. 

 

ENDMETHOD. 

3.8 Context Node BTText 

As mentioned before BTText is a dynamic context node. Only at run time we know 
which text type is assigned and should be used. 

On context node initialization we store reference to context controller because in a later 
step we need access to another context node of this context controller. 

 
 

At event ON_NEW_FOCUS (will be called after initialization or after change of BTADMINH) we need to 
determine the relevant text type and read/get the corresponding BOL/GenIL entity. 

 Check whether relation_name is not build yet and whether context is available. 

 Get popup data from context node POPDATA of context controller 

 Extract process type and action name 

 Determine text type from customizing table ???_CSS_S_C and ???_STAT 

 Build relation name "BTTextH_<TDID>". 

 Get or create entity for relation name (code generated by wizard). 

 

 

 

METHOD on_new_focus. 

 

*------------------------------------------------------------------------- 

*Determine Text Type based on Process Type Source and Target Status 

 

  DATA: 

    lr_popdata         TYPE REF TO cl_bsp_wd_value_node, 

    lv_process_type    TYPE crmt_process_type, 

    lv_action_name     TYPE ppfdtt, 

    lv_tdid            TYPE tdid VALUE ''. 

 

*Relation name is not build yet and Popup data is acessible. 

  IF me->relation_name IS INITIAL AND me->???_owner->typed_context IS BOUND. 

 

*Get Popup Data. 

    lr_popdata ?= me->???_owner->typed_context->popdata->collection_wrapper->get_current( ). 

*Extract Process Type and Action Name. 

    lr_popdata-

>get_property_as_value( EXPORTING iv_attr_name = 'PROCESS_TYPE' IMPORTING ev_result = lv_process_

type ). 

    lr_popdata-

>get_property_as_value( EXPORTING iv_attr_name = 'PPF_ACTION' IMPORTING ev_result = lv_action_nam

e ). 

 

*Get Text Type based on Process Type and Action Name. 

*Source Status and Target Status would be better to have a unique key, but they are not needed si

nce text type should be unique for action name. 

    SELECT SINGLE text_type 

      INTO lv_tdid 

      FROM ???_css_s_c AS g JOIN ???_stat AS s 

             ON g~process_type = s~process_type AND 

                g~stsma = s~stsma AND 

                g~user_status = s~user_status AND 

                g~user_status_new = s~user_status_new 

      WHERE s~process_type = lv_process_type AND 

            s~action = lv_action_name AND 

            text_type <> ''. 

 

    IF lv_tdid IS NOT INITIAL. 

 

*Build relation name. 

      me->relation_name = 'BTTextH_' && lv_tdid. 

 

    ENDIF. 

 

  ENDIF. 

 

*Bind/Create context node if we have a relation name. 

  CHECK me->relation_name IS NOT INITIAL. 

 

*------------------------------------------------------------------------- 

 

  DATA: lv_collection TYPE REF TO if_bol_bo_col, 

        entity        TYPE REF TO cl_crm_bol_entity. 

 

*   get collection of dependent nodes 

  entity ?= focus_bo. 

  TRY. 

      lv_collection = entity->get_related_entities( 

             iv_relation_name = me->relation_name ). 

 

      IF lv_collection IS NOT BOUND OR lv_collection->size( ) = 0. 

        IF entity->is_changeable( ) = abap_true. 

          TRY. 

              entity = entity->create_related_entity( 

               iv_relation_name = me->relation_name ). 

            CATCH cx_crm_genil_model_error cx_crm_genil_duplicate_rel. 

*               should never happen 

          ENDTRY. 

          IF entity IS BOUND. 

            CREATE OBJECT lv_collection TYPE cl_crm_bol_bo_col. 

            lv_collection->add( entity ). 

          ENDIF. 

        ENDIF. 

      ENDIF. 

 

    CATCH cx_crm_genil_model_error. 

*       should never happen 

      EXIT. 

    CATCH cx_sy_ref_is_initial. 

  ENDTRY. 

  me->set_collection( lv_collection ). 

 

 

ENDMETHOD. 

BTTextH is a multi-line textarea. We need to help the Web UI framework to identify it as 
text area. 

method GET_P_CONC_LINES. 

 

  CASE iv_property. 

    WHEN if_bsp_wd_model_setter_getter=>fp_fieldtype. 

      rv_value = cl_bsp_dlc_view_descriptor=>field_type_textarea. 

*    WHEN if_bsp_wd_model_setter_getter=>FP_TEXTAREA_ROWS. 

*      rv_value = '15'. 

  ENDCASE. 

 

endmethod. 

3.9 Inbound Plug IP_POPDATA 

When opening the popup the inbound plug IP_POPDATA of the window will be called. 
The popup data will be provided by the caller. We store these data in context node 
POPDATA of the window. The context node POPDATA is connected with POPDATA of 
the component controller, therefore the data will be synchronized. The implementation 
of both context nodes are equal (same controller class). 

View CSS_UI/SubstatusEF is embedded in the window. Therefore it will be shown 
automatically. The view also have a context node POPDATA which is conntected to the 
component controller, therefore the data will be synchronized as well. 

The context node BTAdminH will be synchronized between component controller and 
view using the same technique. Because of technical reasons the initial synchronization 
will not be done by the inbound plug. It will be done by context node binding described 
in section 3.11. 

 

3.10 Outbound Plug OP_CANCEL / OP_OK 

End-users have the capability to perform the status change or to cancel the status 
change. Actually we have a "Perform Status Change" and a "Cancel" button available. 
Additionally the popup can be closed by clicking the X at the right upper corner of the 
popup window. 

When pressing the button "Perform Status Change", the event EH_OK will be raised. 
This event will check all mandatory fields and raise the outbound plug in case all 
mandatory fields are maintained. The outbound plug will raise the corresponding 
outbound plug of the window. The outbound plug of the window will publish the 
outbound plug and close the popup. The Cancel button will behave similar but without 
mandatory field check. 

Please note that end-users can click the X which will bypass the events and plugs and 
therefore the mandatory field check. We cannot avoid this, but we can check for 
substatus and text type using a consistency check. (You remember this features? :-) 

METHOD eh_onok. 

 

  DATA: 

    lr_controller      TYPE REF TO cl_bsp_wd_appl_controller, 

    lr_msg_srv         TYPE REF TO cl_bsp_wd_message_service. 

 

* cancel save because of mandatory fiels 

  lr_controller ?= comp_controller->m_parent. 

  lr_msg_srv ?= me->view_manager->get_message_service( ). 

  IF cl_crm_uiu_bt_tools=>mandatory_flds_no_save_allowed( ir_controller = lr_controller 

                                                           ir_msg_srv    = lr_msg_srv ) = abap_tr

ue. 

    EXIT. 

  ENDIF. 

 

* call outbound plug 

  me->op_ok( ). 

 

ENDMETHOD. 

method OP_OK. 

 

* leave the ui component via an outbound plug of the current window 

  DATA lr_window_controller TYPE REF TO cl_bsp_wd_window. 

  lr_window_controller ?= me->view_manager->get_window_controller( ). 

  lr_window_controller->call_outbound_plug( iv_outbound_plug   = 'OK' 

                                            iv_data_collection = iv_data_collection ). 

 

endmethod. 

 

 
 

method OP_OK. 

 

*Fire Outbound Plug 

 me->fire_outbound_plug( iv_outbound_plug   = 'OK' 

                          iv_data_collection = iv_data_collection ). 

 

endmethod. 

3.11 Interface, Component Usage and Context Node Binding 

UI component CSS_UI is providing one interface to call the popup. To call the popup 
the InterfaceView CSS_UI/MainWindow needs to be called. Additionally context node 
BTADMINH needs to be bound. 

Note: There is no need to bind POPDATA because it will be done by the inbound plug 
described in section 3.9. In case the caller is not able to bind BTADMINH we could do it 
inside inbound plug of the UI component, because POPDATA also contains the 
transaction guid (this solution is not implemented yet!). However, it is easier and more 
transparent if the caller will do the binding. 

The calling UI component (AIC_CMCR_H and AIC_CMCD_H) needs to define a 
component usage and needs to do the context node binding for BTAdminH. All the rest 
will be done by the ChaRM popup framework. 

 
 

AIC_CMCD_H / AIC_CMCR_H: 

 

Component Usage CU_CSS_UI: 

Component Controller Method WD_USAGE_INITIALIZE: 

 

3.12 BADI: Check Pop-up Start & On Close Event 

Actually the popup should always be shown if configured. That is why the method call 
IF_EX_AIC__POPUP_ENHANCE~ON_CHECK_POP_UP_START always returns true. 

 
 

In case of OK: Nothing to do. 

In case of CANCEL: The popup gets closed with outbound plug "CANCEL" (Cancel 
button) or "" (X on right-upper corner). The action execution will be cancelled by setting 
the returning parameter in method 
IF_EX_AIC__POPUP_ENHANCE~ON_CLOSE_EVENT_BEFORE accordingly. But the 

action is still scheduled and will be executed on next transaction save (without popup). 
To fix this issue we need to de-schedule the action. This is what the method is doing. 

 
METHOD if_ex_aic__popup_enhance~on_close_event_before. 

 

  IF iv_popup_interface->get_fired_outbound_plug( ) <> 'OK'. 

 

*Dont start status change action because popup was cancelled. 

    ev_do_not_start_action = abap_true. 

 

*Popup framework is stopping ppf action processing and transaction save. 

*But the ppf action is still scheduled and will be processed on next save. 

*Therefore we have to de-schedule it. Maybe this will be done by the framework in future!? 

*The code was copied (and adjusted) from CL_EX_AIC_POPUP_EXAMPLE-

>IF_EX_AIC__POPUP_ENHANCE~ON_CLOSE_EVENT_BEFORE. 

 

    DATA: 

      lo_context               TYPE REF TO cl_doc_context_crm_order, 

      lo_manager               TYPE REF TO cl_manager_ppf, 

      lo_context_ppf           TYPE REF TO cl_context_ppf, 

      lo_context_manager       TYPE REF TO cl_context_manager_ppf, 

      lo_current_trigger_list  TYPE REF TO cl_trigger_coll_ppf, 

      lo_trigger_cr            TYPE REF TO cl_trigger_ppf. 

 

*Get context 

    CALL FUNCTION 'CRM_ACTION_CONTEXT_CREATE' 

      EXPORTING 

        iv_header_guid                 = iv_guid 

        iv_object_guid                 = iv_guid 

      IMPORTING 

        ev_context                     = lo_context 

      EXCEPTIONS 

        no_actionprofile_for_proc_type = 1 

        no_actionprofile_for_item_type = 2 

        order_read_failed              = 3 

        OTHERS                         = 4. 

 

* get PPF Manager 

    lo_manager = cl_manager_ppf=>get_instance( ). 

 

    lo_context_ppf ?= lo_context. 

 

* get context manager 

    CALL METHOD lo_manager->get_context_manager 

      EXPORTING 

        io_context         = lo_context_ppf 

      RECEIVING 

        ro_context_manager = lo_context_manager. 

 

* get trigger list 

    lo_current_trigger_list = lo_context_manager->get_triggers( iv_action_type ). 

 

*process every action. 

    IF lo_current_trigger_list->get_number_of_elements( ) > 0. 

      DO. 

        CALL METHOD lo_current_trigger_list->get_next_element 

          RECEIVING 

            ro_element        = lo_trigger_cr 

          EXCEPTIONS 

            element_not_found = 1 

            OTHERS            = 2. 

        IF sy-subrc <> 0. 

          EXIT. " if list has no next element 

        ENDIF. 

 

*Check whether action is unprocessed. 

        IF lo_trigger_cr->get_is_inactiv( ) = space AND lo_trigger_cr-

>get_status( ) = sppf_status_unprocessed. 

 

*Deactivate/Delete it. 

          lo_trigger_cr->if_action_ppf~deactivate( ). 

 

        ENDIF. 

 

      ENDDO. 

    ENDIF. 

  ENDIF. 

 

ENDMETHOD. 

4 Consistency Checks and Clean-Up Actions 

SAP Standard is initializing substatus by Input Readiness Check in UI Component. This 
is a very bad solution. That is why we check and correct substatus after every status 
change. However, there is no easy solution, because... 

BADI ORDER_SAVE cannot be used because method CHECK_BEFORE_SAVE or 
PREPARE are called before status change and CHANGE_BEFORE_UPDATE is called 
after changes are sent to database. 

BADI CRM_ORDER_STATUS cannot be used, because AFTER_CHANGE is not called 
for status reset. (SAP Bug!) 

BADI SOCM_CHECK_CONDITION cannot be used, because consistency checks are 
not executed on reset to previous status. On status change the status change will be 
done. After that all consistency checks will be executed in the defined sequence. If one 
consistency check is triggering a status reset, all following checks will be skipped. After 
reset to the source status, no consistency check will be executed for the source status. 
Therefore it can happen, that the validation check will not be executed. 

Solution 1: Use ChaRM Consistency Check (SOCM_CHECK_CONDITION) for 
maintenance check. 

Solution 2: Use CRM_ORDER_STATUS for validation & correction, but react on all 
status changes (user and system). In case of status cancel an error message should be 
(deleted) and added which causes a system status change (I1030). 

4.1 BADI: Clean-up substatus value 

Substatus can be selected in Details form of Change Request or Change Document. 

Substatus can be selected in Substatus popup if activated. 

However on status change the substatus needs to be removed if invalid. 

 Use case 1: We do a status reset but the substatus was already selected for target 
status. -> Clear. 

 Use case 2: We do a status change and substatus was only valid for source status. -
> Clear 

 Use case 3: We do a status change and substatus was selected before status 
change by substatus popup. -> Keep 

 Use case 4: We do a status change and substatus is valid for source and target 
status. -> Keep 

 Use case 5: We cancel status change by pressing cancel button or closing popup. 
Before cancellation we maintained a substatus. -> Clear [supported by dirty sap 
standard input readiness check solution] 

After Change: 

 Get SERVICE_H using function module CRM_SERVICE_H_READ_OW. 

 Exit if substatus is empty (it is valid in this case). 

 Get possible substatus values using method cl_ai_crm_im_utility=>get_reason_ids. 

 Check whether current value is mentioned in the list. Exit if yes. 

 If not: Clear current substatus using function module 
CRM_SERVICE_H_MAINTAIN_OW. 

 

method IF_EX_CRM_ORDER_STATUS~AFTER_CHANGE. 

 

  DATA: 

    ls_service_h_wrk     TYPE crmt_service_h_wrk, 

    ls_service_h_com     TYPE crmt_service_h_com, 

    lt_input_field_names TYPE crmt_input_field_names_tab, 

    ls_input_field_name  LIKE LINE OF lt_input_field_names, 

    lt_status_reason     TYPE aic_stat_reason_tt. 

 

**We are optimistic. 

*  conditions_ok = abap_true. 

 

*Get current value. 

  CALL FUNCTION 'CRM_SERVICE_H_READ_OW' 

    EXPORTING 

      iv_guid                  = is_status_wrk-guid 

    IMPORTING 

      es_service_h_wrk         = ls_service_h_wrk 

    EXCEPTIONS 

      service_header_not_found = 1 

      OTHERS                   = 2. 

 

*SERVICE_H found. 

  CHECK sy-subrc = 0. 

 

**Maintenance check only. That is easy to do. 

*  IF flt_val = '???_CSS_MAINT'. 

* 

*    IF ls_service_h_wrk-/aicrm/reason_id IS INITIAL. 

* 

**Check failed. 

*      conditions_ok = abap_false. 

* 

*    ENDIF. 

* 

*    EXIT. 

* 

*  ENDIF. 

 

*Reason ID found. 

  CHECK ls_service_h_wrk-/aicrm/reason_id IS NOT INITIAL. 

 

*Get valid status reason. 

  CALL METHOD cl_ai_crm_im_utility=>get_reason_ids 

    EXPORTING 

      iv_guid        = is_status_wrk-guid 

*     iv_status      = 

    IMPORTING 

      et_stat_reason = lt_status_reason. 

 

*Check validity. 

  READ TABLE lt_status_reason 

    TRANSPORTING NO FIELDS 

    WITH KEY reason_id = ls_service_h_wrk-/aicrm/reason_id. 

 

*Current reason is not valid and needs to be deleted. 

  CHECK sy-subrc <> 0. 

 

**Check failed. 

*  conditions_ok = abap_false. 

 

  CLEAR ls_service_h_com. 

  ls_service_h_com-ref_guid = is_status_wrk-guid. 

  ls_service_h_com-ref_kind = 'A'. 

  ls_service_h_com-mode     = ' '. 

  CLEAR ls_service_h_com-/aicrm/reason_id. 

 

  ls_input_field_name-fieldname = '/AICRM/REASON_ID'. 

  APPEND ls_input_field_name TO lt_input_field_names. 

 

*Delete invalid reason. 

  CALL FUNCTION 'CRM_SERVICE_H_MAINTAIN_OW' 

    EXPORTING 

      is_service_h_com     = ls_service_h_com 

    CHANGING 

      ct_input_field_names = lt_input_field_names 

    EXCEPTIONS 

      header_change_error  = 1 

      header_create_error  = 2 

      error_occurred       = 3 

      OTHERS               = 4. 

 

  CHECK sy-subrc <> 0. 

 

endmethod. 

4.2 ChaRM Consistency Check: Substatus Maintained 

As mentioned before, we need a consistency check to ensure that a substatus is 
maintained on status change. Of course the substatus popup is ensuring that a 
substatus is selected. But we are able to bypass this mandatory field check by closing 
the popup. It is also possible to do indirect status changes (i.e. by status reset) where 
the popup is not occuring. To ensure consistency we therefore need to check whether 
substatus is maintained. 

The maintenance check for text type is done by another solution and therefore out of 
scope here. 

Note: Validation check is also implemented as consistency check but not used yet 
because of mentioned restrictions and implemented solution described in section 4.1. 

Assign consistency check "Is substatus maintained" to user status "Withdrawn" and 
"Test Failed but OK to go" of Test Task. 

 

 

 
 

Define Consistency Check "Is Substatus maintained?". Note: check "Is substatus 
valid?" is implemented but not used yet. 

BADI implementation for both consistency checks: 

Is substatus maintained? 

 Get SERVICE_H using function module CRM_SERVICE_H_READ_OW. 

 Return false if substatus is emtpy. otherwise return true. 

 

Is substatus valid? 

 Get SERVICE_H using function module CRM_SERVICE_H_READ_OW. 

 Return true if substatus is empty (it is valid in this case). 

 Get possible substatus values using method cl_ai_crm_im_utility=>get_reason_ids. 

 Check whether current value is mentioned in the list. Return true if yes. 

 if not: Clear current substatus using function module 
CRM_SERVICE_H_MAINTAIN_OW and return false. 

 

METHOD if_ex_socm_check_condition~check_condition. 

 

  DATA: 

    ls_service_h_wrk     TYPE crmt_service_h_wrk, 

    ls_service_h_com     TYPE crmt_service_h_com, 

    lt_input_field_names TYPE crmt_input_field_names_tab, 

    ls_input_field_name  LIKE LINE OF lt_input_field_names, 

    lt_status_reason     TYPE aic_stat_reason_tt. 

 

*We are optimistic. 

  conditions_ok = abap_true. 

 

*Get current value. 

  CALL FUNCTION 'CRM_SERVICE_H_READ_OW' 

    EXPORTING 

      iv_guid                  = hf_instance->change_document_id 

    IMPORTING 

      es_service_h_wrk         = ls_service_h_wrk 

    EXCEPTIONS 

      service_header_not_found = 1 

      OTHERS                   = 2. 

 

*SERVICE_H found. 

  CHECK sy-subrc = 0. 

 

*Maintenance check only. That is easy to do. 

  IF flt_val = '???_CSS_MAINT'. 

 

    IF ls_service_h_wrk-/aicrm/reason_id IS INITIAL. 

 

*Check failed. 

      conditions_ok = abap_false. 

 

    ENDIF. 

 

    EXIT. 

 

  ENDIF. 

 

*Reason ID found. 

  CHECK ls_service_h_wrk-/aicrm/reason_id IS NOT INITIAL. 

 

*Get valid status reason. 

  CALL METHOD cl_ai_crm_im_utility=>get_reason_ids 

    EXPORTING 

      iv_guid        = hf_instance->change_document_id 

*     iv_status      = 

    IMPORTING 

      et_stat_reason = lt_status_reason. 

 

*Check validity. 

  READ TABLE lt_status_reason 

    TRANSPORTING NO FIELDS 

    WITH KEY reason_id = ls_service_h_wrk-/aicrm/reason_id. 

 

*Current reason is not valid and needs to be deleted. 

  CHECK sy-subrc <> 0. 

 

*Check failed. 

  conditions_ok = abap_false. 

 

  CLEAR ls_service_h_com. 

  ls_service_h_com-ref_guid = hf_instance->change_document_id. 

  ls_service_h_com-ref_kind = 'A'. 

  ls_service_h_com-mode     = ' '. 

  CLEAR ls_service_h_com-/aicrm/reason_id. 

 

  ls_input_field_name-fieldname = '/AICRM/REASON_ID'. 

  APPEND ls_input_field_name TO lt_input_field_names. 

 

*Delete invalid reason. 

  CALL FUNCTION 'CRM_SERVICE_H_MAINTAIN_OW' 

    EXPORTING 

      is_service_h_com     = ls_service_h_com 

    CHANGING 

      ct_input_field_names = lt_input_field_names 

    EXCEPTIONS 

      header_change_error  = 1 

      header_create_error  = 2 

      error_occurred       = 3 

      OTHERS               = 4. 

  CHECK sy-subrc <> 0. 

 

ENDMETHOD. 

5 Reporting 

There exist no reporting requirements yet. If such a requirement will be raised in future, 
we can read the substatus from field /AICRM/REASON_ID of table 
CRMD_SERVICE_H. The list of possible substatus (incl. text) can be read from table 
aic_stat_reason and aic_stat_reasont. 


